

ROBUSTNESS IN RANDOM INTERCONNECTIONS GRAPHS

Philippe Flajolet, January 11, 1998

(Based on joint work with Kostas Hatzis, Patras)

Random graphs are usually studied by the probabilistic method that is based on inequalities and on approximations of random variables. Many models that appear in this context can however be subjected to analytic methods. This worksheet explores one such situation using Combstruct, Gfun, and the Maple system. The objective is to characterise the interplay between three parameters of a random graph: its **density** (number of edges), its **robustness** to link failures, and its **connectivity** by short paths.

A triple (Γ, s, t) , where Γ is a graph, s and t are two designated nodes (the source and the target), is said to be l -robust if s and t are connected by at least two edge-disjoint vertices of length exactly l . This definition captures an intuitive notion of robustness to link failures, since s and t will remain connected by "short" paths even in the event of a link (i.e., edge) becoming unavailable in the interconnection graph represented by Γ .

The **random graph model** is that of $G_{n,p}$ where the graph has n vertices and each of the possible $\frac{n(n-1)}{2}$ edges is independently taken with probability p . Under this model, and because of the symmetry it implies with respect to the naming of nodes, it is strictly equivalent to examine robustness properties between a fixed source s and destination t or between a random pair of nodes s, t . For definiteness, we adopt the latter formulation. A graph with very small p will not be robust, while a graph with p close to 1 will have a high number of edge-disjoint paths of small length. The purpose is to evaluate the threshold value p_0 of p from which l -robustness becomes likely. In this worksheet, we focus on the problem of estimating the mean number of edge-disjoint pairs of length l between a random source and a random target, and deduce the associated threshold.

The analysis proceeds in stages:

- **Step 1.** Enumerate the number of pairs of paths of length l connecting nodes 1 and $l+1$ on a line, so that both paths use the same set of nodes 1, 2, ..., $l+1$. This is really a problem of counting so-called **avoiding permutations** defined as cyclic permutations with constraints on their so-called "succession gaps", i.e., the values of the differences between successive values $\sigma_{i+1} - \sigma_i$ that are constrained not to belong to the set $\{-1, 0, 1\}$. This steps itself decomposes into:

Step 1a. Enumerate so-called **templates** that are skeletons of permutations with a **distinguished set of exceptions**.

Step 1b. Transform the counting of templates into enumeration of permutations with a distinguished set of exceptions and conclude by means of the **Principle of Inclusion-Exclusion**, a classical combinatorial principle.

- **Step 2.** Modify the model in order to allow for nodes taken from outside the segment 1, 2, ..., $l+1$. We then count **avoiding paths** that may borrow "outer nodes". This situation models the original random graph problem by allowing nodes taken from the whole pool of nodes that are available from the graph Γ .

- **Step 3.** Return to the original random graph problem and obtain the expected number of edge-disjoint pairs between a source and a destination in a random graph Γ that obeys the $G_{n,p}$ model.

[Comtet, 1974] : L. Comtet, *Advanced Combinatorics*, Reidel, 1974.

[EIS]: N. Sloane and S. Plouffe, *The Encyclopedia of Integer Sequences*, Academic Press, 1995.

This Maple worksheet is based on the current versions of the Maple packages *combstruct* and *gfun* (for version V.4) that can be found under <http://www-rocq.inria.fr/algo/>.

```
[ > restart ;
> with(combstruct) ;
[allstructs, count, draw, finished, gfeqns, gfseries, gfsolve, iterstructs, mark, moment, nextstruct,
  prog_gfeqns, prog_gfseries, prog_gfsolve, variance ]
> with(gfun) ;
[Laplace, algebraicsubs, algeqtodiffeq, algeqtoseries, algfuntoalgeq, borel, cauchyproduct,
  diffeq*diffeq, diffeq+diffeq, diffeqtohomdiffeq, diffeqtorec, guesseqn, guessgf, hadamardproduct,
  holexprtodiffeq, invborel, listtoalgeq, listtodiffeq, listtohypergeom, listtolist, listtoratpoly, listtorec,
  listtoseries, listtoseries/Laplace, listtoseries/egf, listtoseries/lgdegf, listtoseries/lgdogf, listtoseries/ogf,
  listtoseries/revegf, listtoseries/revogf, maxdegcoeff, maxdegeqn, maxordereqn, mindegcoeff,
  mindegeqn, minordereqn, optionsgf, poltodiffeq, poltorec, ratpolytocoef, rec*rec, rec+rec,
  rectodiffeq, rectohomrec, rectoproc, seriestoalgeq, seriestodiffeq, seriestohypergeom, seriestolist,
  seriestoratpoly, seriestorec, seriestoseries ]
```

■ 1. Permutations with constrained succession gaps

The goal of this section is to enumerate permutations of $[1, \dots, n]$ that are cyclic, of the form $[1, \tau_2, \dots, \tau_{n-1}, n]$, and with no succession gap $\tau_{j+1} - \tau_j$ in $\{-1, 0, 1\}$. We call such permutations **avoiding permutations**. In another language, what is the number of paths of length $n-1$ that join 1 to n and have no edge in common with those of the line graph defined by the interval $[1, \dots, n]$?

There are no such paths for $n = 2, 3, 4, 5$. Surprisingly enough, the first nontrivial configurations occur at $n = 6$ (try it!), namely

$\{[1, 4, 2, 5, 3, 6], [1, 3, 5, 2, 4, 6]\}$

and for $n = 7$, one has

$\{[1, 3, 6, 4, 2, 5, 7], [1, 3, 5, 2, 6, 4, 7], [1, 4, 6, 2, 5, 3, 7], [1, 4, 2, 6, 3, 5, 7], [1, 5, 3, 6, 4, 2, 7], [1, 5, 3, 6, 2, 4, 7], [1, 5, 2, 4, 6, 3, 7], [1, 4, 6, 3, 5, 2, 7], [1, 6, 3, 5, 2, 4, 7], [1, 6, 4, 2, 5, 3, 7]\}$

so that the numbers are $Q_6 = 2$, $Q_7 = 10$. A brute force enumeration routine based on the predefined structures available under *Combstruct* is given below.

■ 1.1. Templates

A **template** is a scheme that specifies which "bits and pieces" of the interval $1, 2, \dots, l+1$ may be exceptions. We first need to enumerate templates.

In *combstruct*, we specify templates as decompositions of the interval $1 \dots n$ into *blocks* that are either

- isolated points p ,
- blocks of contiguous unit intervals (based at integer points) oriented left-to-right *LR*,
- blocks of contiguous unit intervals (based at integer points) oriented right-to-left *RL*.

For instance, for $n = 13$, the template

[[1, 2, 3], [4], [5, 6], [7], [8], [11, 10, 9], [12, 13]]

will correspond to any cyclic permutation that has successions of values (in the cycle traversal)

1,2; 2,3; 5,6; 11,10; 10,9; 12,13

as distinguished exceptions to the basic constraint of avoiding permutations.

■ The combinatorial specification

First a preliminary specification. Let $\{a, b\}$ be a binary alphabet. The collection of strings beginning and ending with a letter a is described by

```
[ > sp0 := S = Prod ( Sequence ( Prod ( a , Sequence ( b ) ) ) , a ) ;
      sp0 := S = Prod ( Sequence ( Prod ( a , Sequence ( b ) ) ) , a )
```

(It suffices to decompose according to each occurrence of the letter a .)

Now, the three types of blocks in a template are described by

```
[ > sp1 := Prod ( begin_blockP , Z , end_blockP ) ;
      sp2 := Prod ( begin_blockLR , Z , Sequence ( Prod ( mu_length , Z ) , card >= 1
      ) , end_blockLR ) ;
      sp3 := Prod ( begin_blockRL , Sequence ( Prod ( mu_length , Z ) , card >= 1 ) ,
      Z , end_blockRL ) ;
```

$sp1 := \text{Prod}(\text{begin_blockP}, Z, \text{end_blockP})$

$sp2 := \text{Prod}(\text{begin_blockLR}, Z, \text{Sequence}(\text{Prod}(\text{mu_length}, Z), 1 \leq \text{card}), \text{end_blockLR})$

$sp3 := \text{Prod}(\text{begin_blockRL}, \text{Sequence}(\text{Prod}(\text{mu_length}, Z), 1 \leq \text{card}), Z, \text{end_blockRL})$

Clearly, $sp2$ and $sp3$ are combinatorially isomorphic. For reasons related to application of the inclusion exclusion argument, we keep track of the size (number of nodes $l = 1$ of the basic interval graph) as well as of the length of blocks and of their LR or RL character. Then, the grammar of templates is:

```
[ > Q := subs ( [ a = Union ( sp1 , sp2 ) , b = sp3 ] , sp0 ) , begin_blockP = Epsilon ,
      end_blockP = Epsilon , begin_blockLR = Epsilon ,
      end_blockLR = Epsilon , begin_blockRL = Epsilon ,
      end_blockRL = Epsilon , mu_length = Epsilon ;
```

```
Q := S = Prod ( Sequence ( Prod ( Union ( Prod ( begin_blockP , Z , end_blockP ) ,
      Prod ( begin_blockLR , Z , Sequence ( Prod ( mu_length , Z ) , 1 ≤ card ) , end_blockLR ) ) ,
      Sequence (
      Prod ( begin_blockRL , Sequence ( Prod ( mu_length , Z ) , 1 ≤ card ) , Z , end_blockRL ) ) ) ) ,
      Union ( Prod ( begin_blockP , Z , end_blockP ) ,
      Prod ( begin_blockLR , Z , Sequence ( Prod ( mu_length , Z ) , 1 ≤ card ) , end_blockLR ) ) ) ,
      begin_blockP = E , end_blockP = E , begin_blockLR = E , end_blockLR = E ,
      begin_blockRL = E , end_blockRL = E , mu_length = E
```

```
[ > temp15 := draw ( [ S , { Q } , unlabelled ] , size = 15 ) ;
```

```
temp15 := Prod ( Sequence ( Prod ( %1 , E ) , Prod ( %1 , E ) , Prod ( Prod ( begin_blockLR , Z ,
      Sequence ( Prod ( mu_length , Z ) , Prod ( mu_length , Z ) , Prod ( mu_length , Z ) ) , end_blockLR
      ) , E ) , Prod ( %1 , E ) , Prod ( %1 , E ) , Prod ( %1 , Sequence ( Prod ( begin_blockRL ,
      Sequence ( Prod ( mu_length , Z ) , Prod ( mu_length , Z ) ) , Z , end_blockRL ) ) ) ) ,
      Prod ( Prod ( begin_blockLR , Z , Sequence ( Prod ( mu_length , Z ) ) , end_blockLR ) , E ) ) , %1 )
```

```

> decode:=proc(e) subs([Prod=proc() args end, Sequence=proc()
  args end, mu_length=NULL, Epsilon=NULL], e); ["] end;
decode := proc(e)
  subs([Prod = (proc() args end), Sequence = (proc() args end), mu_length = NULL,
    E = NULL], e);
  ["]
end
> decode(temp15);
[begin_blockP, Z, end_blockP, begin_blockP, Z, end_blockP, begin_blockLR, Z, Z, Z, Z,
  end_blockLR, begin_blockP, Z, end_blockP, begin_blockP, Z, end_blockP, begin_blockP,
  Z, end_blockP, begin_blockRL, Z, Z, Z, end_blockRL, begin_blockLR, Z, Z, end_blockLR,
  begin_blockP, Z, end_blockP]
> seq(count([S, {Q}], unlabelled), size=n), n=0..20);
0, 1, 2, 4, 9, 21, 50, 120, 289, 697, 1682, 4060, 9801, 23661, 57122, 137904, 332929,
  803761, 1940450, 4684660, 11309769

```

Generating functions

A trivariate generating function immediately results from the specification. There, z records size, u records the total number of blocks (needed for subsequent permutation enumerations since blocks should be chained to each other), and v records the total length of LR or RL blocks (the number of distinguished exceptions needed for inclusion exclusion).

```

> gfsolve({Q}, unlabelled, z, [[u, begin_blockP, begin_blockLR, begin_blockRL], [v, mu_length]]);
{S(z, u, v) = -  $\frac{u z (-1 + v z + u z^2 v)}{1 - 2 v z - u z + v^2 z^2 + v^2 z^3 u}$ , end_blockLR(z, u, v) = 1, Z(z, u, v) = z,
  end_blockRL(z, u, v) = 1, begin_blockLR(z, u, v) = u, begin_blockP(z, u, v) = u,
  begin_blockRL(z, u, v) = u, end_blockP(z, u, v) = 1, mu_length(z, u, v) = v}
> f_zuv:=subs(" , S(z, u, v) );

$$f_{zuv} := - \frac{u z (-1 + v z + u z^2 v)}{1 - 2 v z - u z + v^2 z^2 + v^2 z^3 u}$$


```

This GF can be checked by comparing its Taylor expansion and exhaustive listing, at least for small size values.

```

> map(expand, series(f_zuv, z));
u z + (u v + u^2) z^2 + (2 u^2 v + u v^2 + u^3) z^3 + (3 u^2 v^2 + 4 u^3 v + u v^3 + u^4) z^4 +
  (9 u^3 v^2 + 6 u^4 v + 4 u^2 v^3 + u^5 + u v^4) z^5 + O(z^6)

```

The exhaustive listing is given by `Comstruct[allstructs]`

```

> map(decode, allstructs([S, {Q}], unlabelled), size=3);
[[begin_blockLR, Z, Z, end_blockLR, begin_blockP, Z, end_blockP],
  [begin_blockP, Z, end_blockP, begin_blockLR, Z, Z, end_blockLR], [begin_blockP, Z,
  end_blockP, begin_blockP, Z, end_blockP, begin_blockP, Z, end_blockP],
  [begin_blockLR, Z, Z, Z, end_blockLR]]

```

This corresponds to the correct listing

```
[[1],[2,3]]; [[1],[2],[3]]; [1,2,3]; [[1,2],3]
```

hence the monomial $2 u^2 v + u v^2 + u^3$ in the series expansion

1.2. Avoiding permutations

From templates to permutations

Let $F_{n,l}$ be the number of permutations of size n with l distinguished exceptions: $F_{n,l}$ is the number of permutations with l distinguished special successions of the type $j-1, j$ or $j, j-1$. Let Q_n be the number of permutations with no exception. Then, by inclusion exclusion, one has

$$Q_n = \sum_{l=0}^{n-1} F_{n,l} (-1)^l.$$

(Roughly, one takes objects with at least 0 exception, subtract those with at least one exception, then add back those with at least two exceptions, etc.)

Let $F_{n,k,l}$ be the number of templates with a total of n nodes, k blocks, and a total length of the LR and RL blocks equal to l . Then, by looking at all ways to connect the blocks, one has

$$F_{n,l} = \sum_k F_{n,k,l} \phi(k), \text{ and } \phi(1) = 1, \phi(k) = (k-2)! \text{ if } 2 \leq k,$$

since any such connection is determined by an arbitrary permutation of the $k-2$ intermediate blocks.

The trivariate GF of the $F_{n,k,l}$ has been determined in the previous section. This shows a chain by which the Q_n are computable.

Observe that the extension of ϕ by linearity to an arbitrary series in u is given by

$$\phi(h(u)) = \int_0^\infty \frac{e^{(-u)} \left(h(u) - (u-u^2) \left(\frac{\partial}{\partial u} h(u) \right)_{u=0} \right)}{u^2} du.$$

That is to say, we just replace in expansions $u \rightarrow u^2$ and apply the Euler integral

$$\int_0^\infty e^{(-u)} u^k du = k!.$$

Thus, with $F(z, u, v) = \sum F_{n,k,l} z^n u^k v^l$,

the OGF $Q(z) = \sum Q_n z^n$ satisfies $Q(z) = \phi(F(z, u, -1))$

Generating functions

Start from the previously determined trivariate GF and apply the basic chain.

```
> f_zuv; map(expand, series(f_zuv, z));
```

$$-\frac{u z (-1 + v z + u z^2 v)}{1 - 2 v z - u z + v^2 z^2 + v^2 z^3 u}$$

$$u z + (u v + u^2) z^2 + (2 u^2 v + u v^2 + u^3) z^3 + (3 u^2 v^2 + 4 u^3 v + u v^3 + u^4) z^4 +$$

$$(9 u^3 v^2 + 6 u^4 v + 4 u^2 v^3 + u^5 + u v^4) z^5 + O(z^6)$$

```
> f_zu:=subs(v=-1,f_zuv);
```

$$f_{zu} := -\frac{uz(-1-z-uz^2)}{1+2z-uz+z^2+z^3u}$$

Application of the ϕ -transformation (that count the number of ways to connect the blocks) requires the modified form

```
> k_zu:=normal(f_zu-(u-u^2)*subs(u=0,diff(f_zu,u)));
```

$$k_{zu} := \frac{zu^2(uz^2+2z-uz+1)}{(1+z)(uz^2+z-uz+1)}$$

The OGF is here

```
> Q_z:=Int(k_zu*exp(-u)/u^2,u=0..infinity);
```

$$Q_z := \int_0^{\infty} \frac{z(uz^2+2z-uz+1)e^{-u}}{(1+z)(uz^2+z-uz+1)} du$$

This solves the original counting problem numerically:

```
> Q_ser:=series(map(proc(e) int(e*exp(-u)/u^2,u=0..infinity)
end,map(expand,convert(series(k_zu,z=0,15),polynom))),z,15);
```

$$Q_{ser} := z + 2z^6 + 10z^7 + 68z^8 + 500z^9 + 4174z^{10} + 38774z^{11} + 397584z^{12} + 4462848z^{13} + 54455754z^{14}$$

Closed form. The quantity Q_z can be expressed in terms of the exponential integral:

```
> Q_z_closed:=eval(subs(Int=int,Q_z));
```

$$Q_{z_closed} := \frac{\left(z-1 + e^{\left(\frac{1+z}{z(z-1)} \right)} \text{Ei} \left(1, \frac{1+z}{z(z-1)} \right) \right) z}{(z-1)(1+z)}$$

Since one deals with ordinary generating functions (OGF's), this is to be taken as a formal (asymptotic) series:

```
> map(normal,subs(z=1/y,Q_z_closed));map(simplify,asympt(",y,1
1));
```

$$1 - y + e^{\left(\frac{-(y+1)y}{-1+y} \right)} \text{Ei} \left(1, -\frac{(y+1)y}{-1+y} \right) y$$

$$- \frac{(-1+y)(y+1)}{(-1+y)(y+1)}$$

$$\frac{1}{y} + \frac{2}{y^6} + \frac{10}{y^7} + \frac{68}{y^8} + \frac{500}{y^9} + \frac{4174}{y^{10}} + O\left(\frac{1}{y^{11}}\right)$$

The exponential integral (Ei) involves the divergent series of factorials:

```
> Sum(n!*(-y)^(-n-1),n=0..infinity)=asympt(Ei(1,y)*exp(y),y,10
);
```

$$\sum_{n=0}^{\infty} n! (-y)^{(-n-1)} = \frac{1}{y} - \frac{1}{y^2} + \frac{2}{y^3} - \frac{6}{y^4} + \frac{24}{y^5} - \frac{120}{y^6} + \frac{720}{y^7} - \frac{5040}{y^8} + \frac{40320}{y^9} + O\left(\frac{1}{y^{10}}\right)$$

The divergent series is also a hypergeometric series:

```
> series(hypergeom([1,1],[],z),z=0,9);
```

$$1 + z + 2z^2 + 6z^3 + 24z^4 + 120z^5 + 720z^6 + 5040z^7 + 40320z^8 + O(z^9)$$

forms:.

```

> convert_hypergeom:=proc(e) subs(Ei=proc(a,b) if a<>1 then
  ERROR('unable to convert') else
  exp(-b)/b*hypergeom([1,1],[ ],-1/b) fi end,e); simplify("");
end;

convert_hypergeom := proc(e)
  subs(Ei = (proc(a, b)
    if a ≠ 1 then ERROR( 'unable to convert' )
    else exp(-b)*hypergeom([ 1, 1], [ ], - 1 / b) / b
    fi
  end), e);
  simplify("")
end

```

Hence another closed form for the OGF of the Q_n

```

> Q_z_closed2:=convert_hypergeom(Q_z_closed);

```

$$Q_z_closed2 := \frac{z \left(z + z \operatorname{hypergeom} \left([1, 1], [], - \frac{z(z-1)}{1+z} \right) + 1 \right)}{(1+z)^2}$$

Recurrences. Holonomic descriptions (by means of linear differential equations) can be obtained by [Gfun\[holexpertodiffeq\]](#).

```

> Qz_ode0:=holexpertodiffeq(Q_z_closed2, Y(z));

```

$$Qz_ode0 := (z^4 + z^5 + 4z^3 - 1 - z + 4z^2) Y(z) + (-2z^4 + z^2 + z^6) \left(\frac{\partial}{\partial z} Y(z) \right) - z^4 - 2z^3_C_0 - 2z^3 - z^4_C_0 +_C_0 z^2 - z^5 + z - z^2$$

```

> subs(Y(z)=Qz_ser, Qz_ode0):series(" , z=0, 11);
- Qz_ser + (-Qz_ser + 1) z + (4 Qz_ser - 1 + _C_0) z^2 + (4 Qz_ser - 2 _C_0 - 2) z^3 +
(-_C_0 - 1 + Qz_ser) z^4 + (-1 + Qz_ser) z^5
> Qz_ode:=subs(_C[0]=1, Qz_ode0);
Qz_ode :=
(z^4 + z^5 + 4z^3 - 1 - z + 4z^2) Y(z) + (-2z^4 + z^2 + z^6) \left( \frac{\partial}{\partial z} Y(z) \right) - 2z^4 - 4z^3 - z^5 + z

```

The transformation to a linear recurrence is obtained by [Gfun\[diffeqtorec\]](#)

```

> Q_rec:=diffeqtorec({Qz_ode=0, Y(0)=0}, Y(z), u(n));
Q_rec := { u(0) = 0, u(1) = 1,
(n + 1) u(n) + u(n + 1) - 2 n u(n + 2) + 4 u(3 + n) + (3 + n) u(n + 4) - u(n + 5),
u(2) = 0, u(3) = 0, u(4) = 0, u(5) = 0 }

```

This provides an algorithm that uses a linear number of arithmetic operations to determine the quantities Q_n .

```

> ha:=rectoproc(Q_rec, u(n), remember);
ha := proc(n)
option remember;
-4*procname(n - 5) + 4*procname(n - 2) + procname(n - 4)

```

```
+ (-2*procname(n - 3) + procname(n - 5) + procname(n - 1))*n
```

```
end
```

```
> ha(5); ha(50); ha(500);
```

```
0
```

```

1610762789255012501855846297689494046193178343355755998487686
65917861886392480275792026997724027418009221104165776269363463004438380\
55766660072459036707806034973701732314235277671098999360080342295417003\
92144282505597945561328426013937966521561345817045884498867592832897938\
08307184381060210443437630596457794302531018452364381678204788379458561\
63319283244603941468256360854535324043198812649740059680872655870626912\
85454120911586459406436421191277596121471930913837355151842093002557978\
07665388461082629684504192961649653354412434776564136773271656002555317\
91126454540789554091814662127324270713063638200801096363585372704668385\
58068527692374178581063316309789026101221004745226182671038004326069705\
77531265432975469842338524166498415623569253925567794429499540323344624\
33153714048874738680031556218495445663851728355972608489727584438744232\
71017007843907015007416644383573987606586308556317833384896267539628278\
57149740265532256262421765833287015780225404361472631629605832967097105\
49770991557886041758178283805641563298392015790061691730027562959573716\
39199917376529472990059986681882194726437566769717959443857298155265292\
535858523609764515938314672724480762724541633037484152303637096

```

Asymptotics.

An asymptotic pattern is easily inferred from numerical values..

```
> seq(ha(n) / (n-2)! * 1.0, n=2..50);
```

```

0, 0, 0, 0, .083333333333, .083333333333, .094444444444, .09920634921, .1035218254,
.1068507496, .1095634921, .1118037518, .1136859543, .1152893535, .1166716154,
.1178755314, .1189335273, .1198706183, .1207064111, .1214564904, .1221333993,
.1227473475, .1233067294, .1238185103, .1242885187, .1247216685, .1251221319,
.1254934744, .1258387611, .1261606414, .1264614174, .1267430987, .1270074478,
.1272560166, .1274901772, .1277111470, .1279200102, .1281177350, .1283051895,
.1284831534, .1286523298, .1288133538, .1289668009, .1291131935, .1292530072,
.1293866758, .1295145955, .1296371296, .1297546110

```

```
> ha(500) / (498)! * 1.0;
```

```
.1347923134
```

```
> exp(-2) = exp(-2.0);
```

```
e(-2) = .1353352832
```

And one can even guess the next term in an asymptotic expansion

```
> seq(n * (1 - ha(n) / (n-2)! * exp(2.0)), n=6..50);
```

```

2.305471951, 2.689717276, 2.417157614, 2.402628483, 2.350714246, 2.315211987,
2.285150526, 2.260385477, 2.239553484, 2.221807491, 2.206510218, 2.193188459,
2.181482905, 2.171116259, 2.161871138, 2.153574755, 2.146088142, 2.139298156,

```



```

2.105110044, 2.101550239, 2.098223760, 2.095108406, 2.092184669, 2.089435404,
2.086845430, 2.084401301, 2.082090999, 2.079903883, 2.077830280, 2.075861644,
2.073990148, 2.072208808, 2.070511221, 2.068891655, 2.067344825, 2.065865923,
2.064450634, 2.063094868, 2.061795010
> nn:=500; ha(nn)/( (nn-2)!*(1.0-2.0/nn)*exp(-2.0) );
nn := 500
.9999879174

```

Thus, with reasonable certainty, we expect

$$\frac{Q_n}{(n-2)!} = e^{(-2)} \left(1 - \frac{2}{n} + O\left(\frac{1}{n^2}\right) \right)$$

The principle of a proof based on the generating function method is that

$$\text{coeff}_z \text{hypergeom}([1, 1], [], z + dz^2 + O(z^3)) = n! e^d (1 + o(1)),$$

provided that the argument of the hypergeometric is a function that is analytic at the origin.

Here, one has

```

> Q_z_closed2;
z ( z + z hypergeom( [1, 1], [ ], - z(z-1)/(1+z) ) + 1 )
(1+z)^2
> -z*(z-1)/(1+z)=series(-z*(z-1)/(1+z), z=0, 6);
- z(z-1)/(1+z) = z - 2z^2 + 2z^3 - 2z^4 + 2z^5 + O(z^6)

```

so that the asymptotic proportion of legal permutations is *provedly* equal to $e^{(-2)}$:

```

> Qn_asy := 'Q '[n]/(n-2)! = exp(-2)*(1+O(1/n));
Qn_asy := Q_n / (n-2)! = e^{(-2)} (1 + O(1/n))

```

Results

We have thus obtained here a few original results regarding the enumeration of avoiding permutations.

Theorem 1. *The number Q_n of avoiding permutations has ordinary generating function:*

```

> Q_z_closed=Q_z_closed2;
( z - 1 + e^{(1+z)/(z(z-1))} Ei(1, (1+z)/(z(z-1))) ) z
( z + z hypergeom( [1, 1], [ ], - z(z-1)/(1+z) ) + 1 )
( z - 1 ) ( 1 + z ) = ( 1 + z )^2

```

The coefficients Q_n satisfy the recurrence:

```

> Q_rec;
{ u(0) = 0, u(1) = 1,
(n+1)u(n) + u(n+1) - 2nu(n+2) + 4u(3+n) + (3+n)u(n+4) - u(n+5), u(2) = 0,
u(3) = 0, u(4) = 0, u(5) = 0 }

```

Accordingly, the ordinary generating function satisfies the corresponding linear ODE:

$$(1+z)(z^4+4z^2-1)Y(z)+z^2(z-1)^2(1+z)^2\left(\frac{\partial}{\partial z}Y(z)\right)-2z^4-4z^3-z^5+z$$

The coefficients Q_n satisfy the asymptotic estimate

> Qn_asym;

$$\frac{Q_n}{(n-2)!} = e^{(-2)} \left(1 + O\left(\frac{1}{n}\right) \right)$$

2. Paths with outer nodes

We now define **avoiding paths**. An avoiding path of size n is a sequence of values from the set $\{1 \dots n\} \cup \{-1\}$, where -1 represents generically a node that does not belong to the interval $[1, \dots, n]$, with the constraint that the initial value of the sequence is 1, the final value is n , no successive value pairs can be of the type $j, j+1$ or $j, j-1$, and no value from $[1, \dots, n]$ can occur more than once.

Here are the sets of avoiding paths for sizes $n = 2, 3, 4, 5$:

$$2, \{ \}$$

$$3, \{ [1, -1, 3] \}$$

$$4, \{ [1, -1, -1, 4], [1, -1, 2, 4], [1, 3, -1, 4] \}$$

$$5, \{ [1, 4, 2, -1, 5], [1, -1, 4, 2, 5], [1, 4, -1, 2, 5], [1, 4, -1, 3, 5], [1, 3, -1, 2, 5], [1, -1, -1, 2, 5], [1, 4, -1, -1, 5], [1, -1, -1, -1, 5], [1, -1, -1, 3, 5], [1, 3, -1, -1, 5], [1, -1, 2, -1, 5], [1, -1, 4, -1, 5], [1, -1, 3, -1, 5] \}$$

An avoiding path thus describes a path from 1 to n that is allowed to go through some "outer nodes". The goal is here to enumerate avoiding paths by size and by number of outer nodes conventionally represented by -1 .

Let $Q_{n,j}$ be the number of avoiding paths of size n with j outer nodes. Clearly, $Q_{n,0} = Q_n$ is the number of avoiding permutations. An extension of the earlier argument is thus needed.

2.1. Templates

One first needs templates on which an inclusion-exclusion argument is applied. The specifications are a simple modification of templates associated to avoiding permutations.

Let $\{a, b, x\}$ be a ternary alphabet. The collection of strings beginning with a and containing only one x that occurs at the end is described by

$$\begin{aligned} > \text{sp0} := S = \text{Prod}(\text{Sequence}(\text{Prod}(a, \text{Sequence}(b))), x); \\ & \text{sp0} := S = \text{Prod}(\text{Sequence}(\text{Prod}(a, \text{Sequence}(b))), x) \end{aligned}$$

(It suffices to decompose according to each occurrence of the letter a .)

We first need so-called "outer points" that are taken from outer space.

$$\begin{aligned} > \text{Outerpoints} := \text{Sequence}(\text{Prod}(Z, \text{mu_outerpoint})); \\ & \text{Outerpoints} := \text{Sequence}(\text{Prod}(Z, \text{mu_outerpoint})) \end{aligned}$$

We also need "inner points".

$$\begin{aligned} > \text{Innerpoints} := \text{Sequence}(\text{Prod}(Z, \text{mu_innerpoint})); \\ & \text{Innerpoints} := \text{Sequence}(\text{Prod}(Z, \text{mu_innerpoint})) \end{aligned}$$

Size is defined as the cumulated number of points in the pair of paths that underlies an avoiding path in the sense above: it is thus equal to the length of the avoiding path plus the number of -1 symbols corresponding to the outer nodes. We thus introduce a special notation for nodes of the integer line that are shared by the two paths:

$$Z2 := \text{Prod}(Z, Z)$$

Now, the three types of blocks are described by

```
> sp1:=Prod(mu_block, Z2, Outerpoints, Innerpoints);
   sp2:=Prod(mu_block, Z2, Sequence(Prod(mu_length, Z2), card>=1), Outerpoints, Innerpoints);
   sp3:=Prod(Sequence(Prod(mu_length, Z2), card>=1), Z2, Outerpoints, Innerpoints, mu_block);
```

```
sp1 := Prod(mu_block, Prod(Z, Z), Sequence(Prod(Z, mu_outerpoint),
      Sequence(Prod(Z, mu_innerpoint))))
```

```
sp2 := Prod(mu_block, Prod(Z, Z), Sequence(Prod(mu_length, Prod(Z, Z)), 1 ≤ card),
      Sequence(Prod(Z, mu_outerpoint)), Sequence(Prod(Z, mu_innerpoint)))
```

```
sp3 := Prod(Sequence(Prod(mu_length, Prod(Z, Z)), 1 ≤ card), Prod(Z, Z),
      Sequence(Prod(Z, mu_outerpoint)), Sequence(Prod(Z, mu_innerpoint)), mu_block)
```

(Clearly, $sp2$ and $sp3$ are combinatorially isomorphic.)

The blocks that can occur at the end are of type x and can only be of type $sp1$ or $sp2$ but without outer points nor innerpoints.

```
> sp1x:=Prod(mu_block, Z2);
      sp1x := Prod(mu_block, Prod(Z, Z))
> sp2x:=Prod(mu_block, Z2, Sequence(Prod(mu_length, Z2), card>=1));
      sp2x := Prod(mu_block, Prod(Z, Z), Sequence(Prod(mu_length, Prod(Z, Z)), 1 ≤ card))
```

Then, the grammar is:

```
> Q:=subs([a=Union(sp1, sp2), b=sp3, x=Union(sp1x, sp2x)], sp0),
   mu_block=Epsilon,
   mu_length=Epsilon, mu_outerpoint=Epsilon, mu_innerpoint=Epsilon;
Q := S = Prod(Sequence(Prod(Union(Prod(mu_block, Prod(Z, Z),
      Sequence(Prod(Z, mu_outerpoint)), Sequence(Prod(Z, mu_innerpoint))), Prod(mu_block,
      Prod(Z, Z), %1, Sequence(Prod(Z, mu_outerpoint)), Sequence(Prod(Z, mu_innerpoint))),
      Sequence(Prod(%1, Prod(Z, Z), Sequence(Prod(Z, mu_outerpoint)),
      Sequence(Prod(Z, mu_innerpoint))), mu_block))),
      Union(Prod(mu_block, Prod(Z, Z)), Prod(mu_block, Prod(Z, Z), %1))), mu_block = E,
      mu_length = E, mu_outerpoint = E, mu_innerpoint = E
      %1 := Sequence(Prod(mu_length, Prod(Z, Z)), 1 ≤ card)
> draw([S, {Q}, unlabelled], size=8);
Prod(Sequence(Prod(Prod(mu_block, Prod(Z, Z), E, E), E),
      Prod(Prod(mu_block, Prod(Z, Z), E, E), E)),
      Prod(mu_block, Prod(Z, Z), Sequence(Prod(mu_length, Prod(Z, Z))))))
> seq(count([S, {Q}, unlabelled], size=n), n=0..20);
0, 0, 1, 0, 2, 2, 7, 12, 30, 62, 143, 312, 702, 1554, 3471, 7716, 17198, 38278, 85263, 189840,
422782
> for j from 0 to 6 do allstructs([S, {Q}, unlabelled], size=j) od;
```

[]

[]

[

```
Prod(Sequence(Prod(Prod(mu_block, Prod(Z, Z), E, E), E)), Prod(mu_block, Prod(Z, Z))),
Prod(E, Prod(mu_block, Prod(Z, Z), Sequence(Prod(mu_length, Prod(Z, Z)))))])
```

[Prod(

```
Sequence(Prod(Prod(mu_block, Prod(Z, Z), E, Sequence(Prod(Z, mu_innerpoint))), E)),
Prod(mu_block, Prod(Z, Z))), Prod(
Sequence(Prod(Prod(mu_block, Prod(Z, Z), Sequence(Prod(Z, mu_outerpoint))), E), E)),
Prod(mu_block, Prod(Z, Z)))]
```

[Prod(Sequence(Prod(Prod(mu_block, Prod(Z, Z), Sequence(Prod(Z, mu_outerpoint))),

```
Sequence(Prod(Z, mu_innerpoint))), E)), %2), Prod(Sequence(Prod(Prod(mu_block,
Prod(Z, Z), E, Sequence(Prod(Z, mu_innerpoint), Prod(Z, mu_innerpoint))), E)), %2),
Prod(Sequence(Prod(Prod(mu_block, Prod(Z, Z),
Sequence(Prod(Z, mu_outerpoint), Prod(Z, mu_outerpoint))), E), E)), %2),
Prod(Sequence(%3), Prod(mu_block, Prod(Z, Z), Sequence(%1))),
Prod(Sequence(%3, %3), %2),
```

```
Prod(Sequence(Prod(Prod(mu_block, Prod(Z, Z), Sequence(%1), E, E), E)), %2),
Prod(E, Prod(mu_block, Prod(Z, Z), Sequence(%1, %1)))]
```

```
%1 := Prod(mu_length, Prod(Z, Z))
```

```
%2 := Prod(mu_block, Prod(Z, Z))
```

```
%3 := Prod(Prod(mu_block, Prod(Z, Z), E, E), E)
```

2.2. Generating functions

The trivariate generating function immediately results from the specification.

```
> gfsolve({Q}, unlabelled, z, [[u, mu_block], [v, mu_length], [w1, mu_out
erpoint], [w2, mu_innerpoint]]);
{ mu_innerpoint(z, u, v, w1, w2) = w2, Z(z, u, v, w1, w2) = z, mu_block(z, u, v, w1, w2) = u,
mu_outerpoint(z, u, v, w1, w2) = w1, mu_length(z, u, v, w1, w2) = v, S(z, u, v, w1, w2) = -u
z^2 (-1 + z w2 + z w1 - z^2 w1 w2 + v z^2 - v z^3 w2 - v z^3 w1 + v z^4 w1 w2 + u z^4 v) / (1 - z w2
- z w1 + z^2 w1 w2 - 2 v z^2 + 2 v z^3 w2 + 2 v z^3 w1 - 2 v z^4 w1 w2 - u z^2 + v^2 z^4 - v^2 z^5 w2
- v^2 z^5 w1 + v^2 z^6 w1 w2 + v^2 z^6 u) }
```

For inclusion-exclusion, one must set $v=-1$:

```
> f_zu := factor(subs(v=-1, subs(", S(z, u, v, w1, w2)))));
f_zu := \frac{u^2 z^2 (1 - z w2 - z w1 + z^2 w1 w2 + z^2 - z^3 w2 - z^3 w1 + z^4 w1 w2 + u z^4)}{(1 + z^2) (z^4 w1 w2 + u z^4 - z^3 w2 - z^3 w1 + z^2 w1 w2 + z^2 - u z^2 - z w2 - z w1 + 1)}
```

Application of the ϕ -transformation (that counts the number of ways to connect the blocks) requires the modified form

```
> h_zu := factor(normal(f_zu - (u - u^2) * subs(u=0, diff(f_zu, u)))));
h_zu := \frac{u^2 z^2 (1 + 2 z^2 - z^3 w1 + u z^4 + z^4 w1 w2 - z^3 w2 - u z^2 - z w2 - z w1 + z^2 w1 w2)}{(1 + z^2) (z^4 w1 w2 + u z^4 - z^3 w2 - z^3 w1 + z^2 w1 w2 + z^2 - u z^2 - z w2 - z w1 + 1)}
```

```

> Q_ser:=series(map(proc(e) int(e*exp(-u)/u^2,u=0..infinity)
end,map(expand(convert(series(h_zu,z=0,12),polynom))),z,15);
Q_ser:=z^2+(w2+w1)z^5+(w1^2+w2w1+w2^2)z^6+(w1^3+w2^2w1+w2w1^2+w2^3)z^7+
(w1^4+w2^2+w2^2w1^2+2w2w1+w1^3w2+w1^2+w2^4+w2^3w1)z^8+(4w2w1^2+w2^4w1
+w2^2w1^3+2w2^3+w2^3w1^2+w1+2w1^3+w1^5+w1^4w2+w2^5+4w2^2w1+w2)z^9+(
6w2^3w1+6w1^3w2+w2^5w1+5w2w1+w1^6+3w1^4+7w2^2w1^2+w2^4w1^2+3w2^4
+3w1^2+3w2^2+w1^5w2+w1^4w2^2+w2^3w1^3+w2^6)z^10+(10w2^2w1^3+w1^6w2+w2^6w1
+2w2+4w1^5+4w2^5+7w1^3+w2^4w1^3+w1^7+15w2w1^2+7w2^3+15w2^2w1
+8w1^4w2+w2^3w1^4+w2^7+10w2^3w1^2+2w1+w2^5w1^2+8w2^4w1+w1^5w2^2)z^11

```

There, the coefficient of $z^n w1^j w2^j$ counts the number of avoiding paths with size n and with j outer nodes.

Closed form and recurrences. The OGF is here

```

> Q_z:=Int(h_zu*exp(-u)/u^2,u=0..infinity);
Q_z:=

```

$$\int_0^{\infty} \frac{z^2(1+2z^2-z^3w1+uz^4+z^4w1w2-z^3w2-uz^2-zw2-zw1+z^2w1w2)e^{-u}}{(1+z^2)(z^4w1w2+uz^4-z^3w2-z^3w1+z^2w1w2+z^2-uz^2-zw2-zw1+1)} du$$

And this can be expressed in terms of the exponential integral:

```

> Q_z_closed:=eval(subs(Int=int,Q_z));

```

$$Q_{z_closed} := z^2 \left(z^2 e^{\left(\frac{(1+z^2)(w2+w1)}{z(z-1)(1+z)} \right)} + \operatorname{Ei} \left(1, \frac{z^4 w1 w2 - z^3 w2 - z^3 w1 + z^2 w1 w2 + z^2 - z w2 - z w1 + 1}{z^2 (z^2 - 1)} \right) e^{\left(\frac{(1+z^2)(z^2 w1 w2 + 1)}{z^2 (z-1)(1+z)} \right)} - e^{\left(\frac{(1+z^2)(w2+w1)}{z(z-1)(1+z)} \right)} \right) / ((z^2 - 1)(1 + z^2))$$

Again, there is an "explicit form" of the OGF the problem

```

> Qz_closed2:=map(factor,convert_hypergeom(Q_z_closed));
Qz_closed2:=

```

$$z^4 w1 w2 - z^3 w2 - z^3 w1 + z^2 \operatorname{hypergeom} \left([1, 1], [], - \frac{z^2 (z-1)(1+z)}{(1+z^2)(z w2 - 1)(z w1 - 1)} \right) + z^2 + z^2 w1 w2 - z w2 - z w1 + 1 \Big) z^2 / ((1+z^2)^2 (z w2 - 1)(z w1 - 1))$$

```

> Qz_closed3:=factor(Qz_closed2-subst(hypergeom=0,Qz_closed2))+factor(subst(hypergeom=0,Qz_closed2));

```

$$Qz_closed3 := \frac{z^4 \text{hypergeom}\left([1, 1], [1, -], \frac{(1+z^2)(zw2-1)(zw1-1)}{(1+z^2)^2(zw2-1)(zw1-1)}\right)}{(1+z^2)^2(zw2-1)(zw1-1)} + \frac{z^2}{1+z^2}$$

The diagonal, where $w1$ and $w2$ appear with equal exponents, is then easily extracted.

```
> Q_order:=20;
Qz_ser:=map(normal, series(Qz_closed3, z=0, Q_order+2)) : subs([w1=w, w2=t/w], Qz_ser) : subs([w1=w, w2=t/w], Qz_ser) :
map(series, "", w=0, 2*Q_order+5) : Qzt_ser:=map(coeff, "", w, 0);
Qzt_ser := z^2 + t z^6 + (t^2 + 2 t) z^8 + (t^3 + 5 t + 7 t^2) z^10 + (20 t + 39 t^2 + t^4 + 14 t^3 + 2) z^12 +
(139 t^3 + 115 t + t^5 + 23 t^4 + 10 + 245 t^2) z^14 +
(353 t^4 + 790 t + 1340 t^3 + t^6 + 1829 t^2 + 34 t^5 + 68) z^16 +
(741 t^5 + t^7 + 4871 t^4 + 15771 t^2 + 47 t^6 + 13749 t^3 + 500 + 6217 t) z^18 +
(153114 t^3 + 1375 t^6 + 55160 t + 62 t^7 + t^8 + 13844 t^5 + 66955 t^4 + 4174 + 153043 t^2) z^20 +
O(z^22)
```

We thus get results that generalize the counting of avoiding permutations.

```
> subs(t=0, Qzt_ser) ; subs(t=1, Qzt_ser) ;

z^2 + 2 z^12 + 10 z^14 + 68 z^16 + 500 z^18 + 4174 z^20 + O(z^22)
z^2 + z^6 + 3 z^8 + 13 z^10 + 76 z^12 + 533 z^14 + 4415 z^16 + 41897 z^18 + 447728 z^20 + O(z^22)
```

2.3. Exhaustive listings

Here's a procedure that lists balanced avoiding pairs exhaustively (by generating all permutations and filtering the relevant ones) and derives the counting polynomials. This fully confirms the GF computations done previously.

```
> test := proc()
local i, perms, dontcares, poly, count1, p, p1, j, x, s, n,
listing; n:=args[1];
perms:=combstruct[allstructs](Permutation([seq(i, i=2..n-1)]), size=n-2); dontcares :=
combstruct[allstructs](Subset({seq(i, i=2..n-1)}));
listing:={};
poly:=0;
for s in dontcares do
count1:=0;
for p in perms do p1:=[1, op(p), n];
for x in s do p1[x]:=-1; od;
for j from 1 to n-1 while abs(p1[j+1]-p1[j])>1 do
od;
if j=n then count1:=count1+1; listing:=listing union {p1} fi;
od;
poly:=poly+count1*t^nops(s)/nops(s)!;
od;
if nargs>1 then RETURN(sort(listing)) fi;
RETURN(poly);
end;
```

test := **proc**()

local *i, perms, dontcares, poly, count1, p, p1, j, x, s, n, listing;*

```

perms := combstruct[allstructs](Permutation([seq(i, i = 2 .. n - 1)]), size = n - 2);
dontcares := combstruct[allstructs](Subset({seq(i, i = 2 .. n - 1)}));
listing := { };
poly := 0;
for s in dontcares do
  count1 := 0;
  for p in perms do
    p1 := [1, op(p), n];
    for x in s do p1[x] := -1 od;
    for j to n - 1 while abs(p1[j + 1] - p1[j]) ≠ 1 do od;
    if j = n then count1 := count1 + 1; listing := listing union {p1} fi
  od;
  poly := poly + count1 * t^nops(s) / nops(s)!
od;
if 1 < nargs then RETURN(sort(listing)) fi;
RETURN(poly)

```

end

```

> for j from 2 to 6 do j, test(j) od;
      2, 0
      3, t
      4, t2 + 2 t
      5, t3 + 5 t + 7 t2
      6, 20 t + 39 t2 + t4 + 14 t3 + 2
> for j from 2 to 6 do j, test(j, LIST_THEM_ALL) od;
      2, { }
      3, {[1, -1, 3]}
      4, {[1, 3, -1, 4], [1, -1, -1, 4], [1, -1, 2, 4]}
5, {[1, -1, -1, 2, 5], [1, 3, -1, 2, 5], [1, -1, -1, 3, 5], [1, -1, 4, 2, 5], [1, 4, -1, 2, 5],
   [1, 4, -1, 3, 5], [1, -1, -1, -1, 5], [1, 4, -1, -1, 5], [1, 3, -1, -1, 5], [1, 4, 2, -1, 5],
   [1, -1, 2, -1, 5], [1, -1, 4, -1, 5], [1, -1, 3, -1, 5]}
6, {[1, 3, -1, -1, 4, 6], [1, 3, -1, -1, 2, 6], [1, 4, -1, -1, 3, 6], [1, 4, -1, -1, 2, 6],
   [1, 5, -1, -1, 2, 6], [1, 5, -1, -1, 3, 6], [1, 5, -1, -1, 4, 6], [1, -1, 3, -1, 4, 6], [1, -1, 4, -1, 3, 6],
   [1, -1, 5, -1, 3, 6], [1, -1, 5, -1, 4, 6], [1, -1, 4, -1, 2, 6], [1, -1, 2, -1, 4, 6], [1, -1, 5, -1, 2, 6],
   [1, -1, 2, -1, 3, 6], [1, -1, 3, -1, 2, 6], [1, 5, -1, 3, -1, 6], [1, 3, 5, 2, -1, 6], [1, 4, 2, 5, -1, 6],
   [1, 5, 2, 4, -1, 6], [1, 3, -1, 5, -1, 6], [1, 3, -1, 4, -1, 6], [1, 4, -1, 2, -1, 6], [1, 4, -1, 5, -1, 6],
   [1, 4, -1, 3, -1, 6], [1, 3, -1, 2, -1, 6], [1, -1, 3, 5, -1, 6], [1, 5, -1, 2, -1, 6], [1, 5, -1, 4, -1, 6],
   [1, -1, 5, 2, -1, 6], [1, -1, 4, 2, -1, 6], [1, -1, 2, 5, -1, 6], [1, -1, 2, 4, -1, 6], [1, -1, 5, 3, -1, 6],
   [1, -1, -1, 3, -1, 6], [1, -1, -1, 5, -1, 6], [1, -1, -1, 4, -1, 6], [1, -1, -1, 2, -1, 6],
   [1, 3, 5, -1, -1, 6], [1, 5, 2, -1, -1, 6], [1, 4, 2, -1, -1, 6], [1, 5, 3, -1, -1, 6], [1, 4, -1, -1, -1, 6],
   [1, 3, -1, -1, -1, 6], [1, 5, -1, -1, -1, 6], [1, -1, 5, -1, -1, 6], [1, -1, 2, -1, -1, 6],

```

```

[1, -1, 5, 2, 4, 6], [1, -1, -1, -1, -1, 6], [1, -1, 3, 5, 2, 6], [1, -1, -1, 5, 3, 6], [1, 3, -1, 4, 2, 6],
[1, -1, -1, 4, 2, 6], [1, -1, -1, 2, 4, 6], [1, 3, -1, 2, 4, 6], [1, 3, -1, 5, 2, 6], [1, -1, -1, 5, 2, 6],
[1, 5, -1, 4, 2, 6], [1, 5, -1, 2, 4, 6], [1, 4, -1, 5, 2, 6], [1, 4, -1, 5, 3, 6], [1, -1, -1, -1, 3, 6],
[1, -1, -1, -1, 4, 6], [1, -1, -1, -1, 2, 6], [1, 4, 2, -1, 3, 6], [1, 3, 5, -1, 2, 6], [1, 3, 5, -1, 4, 6],
[1, 5, 2, -1, 4, 6], [1, 5, 3, -1, 2, 6], [1, 5, 3, -1, 4, 6], [1, 5, 2, -1, 3, 6]
]
> remove(has, test(6, LIST_THEM_ALL), -1);
      {[1, 4, 2, 5, 3, 6], [1, 3, 5, 2, 4, 6]}
> remove(has, test(7, LIST_THEM_ALL), -1);
{[1, 3, 5, 2, 6, 4, 7], [1, 3, 6, 4, 2, 5, 7], [1, 4, 2, 6, 3, 5, 7], [1, 4, 6, 2, 5, 3, 7],
 [1, 4, 6, 3, 5, 2, 7], [1, 5, 2, 4, 6, 3, 7], [1, 5, 3, 6, 2, 4, 7], [1, 5, 3, 6, 4, 2, 7],
 [1, 6, 3, 5, 2, 4, 7], [1, 6, 4, 2, 5, 3, 7]}

```

These results confirm the GF computation of the previous subsection.

2.4. Explicit binomial formulae

We can now return to the GF of avoiding paths enumerated by size and number of outer nodes.

```

> Qz_closed3;

```

$$\frac{z^4 \operatorname{hypergeom}\left([1, 1], [], -\frac{z^2(z-1)(1+z)}{(1+z^2)(zw^2-1)(zw-1)}\right)}{(1+z^2)^2(zw^2-1)(zw-1)} + \frac{z^2}{1+z^2}$$

The coefficient $c(n, j, k)$ of $z^n w^j w^k$ is obtained by straight expansion and avoiding paths are then enumerated by $C(n, j) = c(n, j, j)$. The corresponding formulae are obtained directly by symbolic expansions (performed manually, though!)

```

> j := 'j' : C_formula := C(n+2, j) = Sum(Sum(
  (-1)^(k1+k2) * (n-j-k1-k2)! * 'B' (n-j-k1-k2, k1) *
  'B' (n-j-k1+1, k2) * 'B' (n-k1-k2, j)^2, k1=0..n-j-k2), k2=0..n-j);

```

$$C_formula := C(n+2, j) = \sum_{k_2=0}^{n-j} \left(\sum_{k_1=0}^{n-j-k_2} (-1)^{(k_1+k_2)} (n-j-k_1-k_2)! B(n-j-k_1-k_2, k_1) B(n-j-k_1+1, k_2) B(n-k_1-k_2, j)^2 \right)$$

where $B(a, b)$ is the binomial coefficient $B(a, b) = \frac{a!}{b!(a-b)!}$.

```

> Bin := proc(n, k) if n < 0 or k < 0 then 0 else binomial(n, k) fi; end;
      Bin := proc(n, k) if n < 0 or k < 0 then 0 else binomial(n, k) fi end
> coef := proc(n, j) option remember; local k1, k2;
  add( add( (-1)^(k1+k2) * (n-j-k1-k2)! * Bin(n-j-k1-k2, k1) *
  Bin(n-j-k1+1, k2) * Bin(n-k1-k2, j)^2, k1=0..n-j-k2), k2=0..n-j);
  end;
coef := proc(n, j)
local k1, k2;
option remember;
  add(add((-1)^(k1+k2)*(n-j-k1-k2)!*Bin(n-j-k1-k2, k1)*
  Bin(n-j-k1+1, k2)*Bin(n-k1-k2, j)^2, k1=0..n-j-k2), k2=0..n-j)

```



```

> Coef:=proc(n,j) local r; option remember; r:=coef(n-2,j);
  if j=0 then r:=r-(-1)^n fi; r; end;

```

Coef:=

```

proc(n,j) local r; option remember; r := coef(n-2,j); if j=0 then r := r - (-1)^n fi; r end

```

This gives the generating polynomials $co(n, t) := \sum_{j=0}^n C(n, j) t^j$.

```

> co:=proc(n,t) local j; option remember;
  add(Coef(n,j)*t^j, j=0..n) end;

```

```

  co := proc(n,t) local j; option remember; add(Coef(n,j)*t^j, j=0..n) end

```

Here is a short table, again consistent with values obtained by exhaustive listing of small cases.

```

> for nn from 2 to 8 do 'co'(nn,t)=co(nn,t) od;

```

$$co(2, t) = 0$$

$$co(3, t) = t$$

$$co(4, t) = t^2 + 2t$$

$$co(5, t) = t^3 + 5t + 7t^2$$

$$co(6, t) = 20t + 39t^2 + t^4 + 14t^3 + 2$$

$$co(7, t) = 139t^3 + 115t + t^5 + 23t^4 + 10 + 245t^2$$

$$co(8, t) = 353t^4 + 790t + 1340t^3 + t^6 + 1829t^2 + 34t^5 + 68$$

Results

Theorem 2. *The number $C_{n,j}$ of avoiding paths counted by size n and by number of outer nodes j satisfies for $0 \leq j$*

```

> C_formula;

```

$$C(n+2, j) = \sum_{k2=0}^{n-j} \left(\sum_{k1=0}^{n-j-k2} \right)$$

$$(-1)^{(k1+k2)} (n-j-k1-k2)! B(n-j-k1-k2, k1) B(n-j-k1+1, k2) B(n-k1-k2, j)^2$$

For $j=0$, the formula simplifies and provides the number of avoiding permutations

```

> 'Q[n+2]' = subs(j=0, B(n-k1-k2, 0)=1, op(2, "))-(-1)^n;

```

$$Q_{n+2} = \left(\sum_{k2=0}^n \left(\sum_{k1=0}^{n-k2} (-1)^{(k1+k2)} (n-k1-k2)! B(n-k1-k2, k1) B(n+1-k1, k2) \right) \right) - (-1)^n$$

(B denotes a binomial coefficient.)

3. The random graph problem

We conclude this section by showing how to estimate the robustness of connections to link failures between a source and a target in a random graph that obeys the $G_{n,p}$ model, where edges between n nodes are chosen independently with probability p .

An **avoiding pair** of length l in a graph is an unordered pair of paths, each of length l that may share some nodes but are edge disjoint. The mean number of (unordered) avoiding pairs in a random graph

robustness of the graph to a single link failure. This mean value is:

$$\begin{aligned}
 &> l := 'l' : \\
 &\text{Numpath_formula} := 1/2 * 1 / (n * (n-1)) * \text{Sum}(C(l+1, j) * B(n, l+1+j) * (l+1+j)!, \\
 &j=0..l) * p^{(2*l)} ; \\
 &\text{Numpath_formula} := \frac{1}{2} \frac{\left(\sum_{j=0}^l C(l+1, j) B(n, l+1+j) (l+1+j)! \right) p^{(2l)}}{n(n-1)}
 \end{aligned}$$

The argument is as follows. The coefficient $\frac{1}{2}$ corresponds to the fact that one takes unordered pairs of

paths, the coefficient $\frac{1}{n(n-1)}$ averages over all sources and destinations, and the arrangement numbers

account for the number of ways to embed an avoiding path into a graph by choosing certain nodes and assigning them in some order to an avoiding path.

The C -coefficients are as provided by Theorem 2 and B still means binomial coefficient.

The procedure that implements the formula for the mean number of paths is then

```

> means:=proc(n,p,l) local j; option remember;
  1/2*1/(n*(n-1))*add(Coef(l+1,j)*binomial(n,l+1+j)*(l+1+j)!,j=0..l+
  1)*p^(2*l); expand(""); if l<=12 then factor("") else " fi end;
means := proc(n, p, l)
local j;
option remember;
  1 / 2*add(Coef(l + 1, j)*binomial(n, l + 1 + j)*(l + 1 + j)!, j = 0 .. l + 1)*p^(2*l) / (n*
  (n - 1));
  expand("");
  if l ≤ 12 then factor("") else " fi
end
> for ll from 1 to 10 do means(n,p,ll) od;
0
1/2 (n - 2) (n - 3) p^4
1/2 (n - 2) (n - 4) (n - 3)^2 p^6
1/2 (n - 1) (n - 2) (n - 3) (n - 4) (n - 5)^2 p^8
1/2 (n - 2) (n - 3) (n - 4) (n^3 - 11 n^2 + 25 n + 32) (n - 5)^2 p^10
1/2 (n - 2) (n - 3) (n - 4) (n - 5) (n - 6) (n^5 - 22 n^4 + 162 n^3 - 333 n^2 - 939 n + 3349) p^12
1/2 (n - 2) (n - 3) (n - 4) (n - 5) (n - 6) (n - 7)
(n^6 - 29 n^5 + 298 n^4 - 1029 n^3 - 2430 n^2 + 23675 n - 39444) p^14

```

$$\frac{1}{2} (n-2)(n-3)(n-4)(n-5)(n-6)(n-7)(n-8)$$

$$(n^7 - 37n^6 + 508n^5 - 2699n^4 - 2984n^3 + 92457n^2 - 343397n + 368987)p^{16}$$

$$\frac{1}{2} (n-2)(n-3)(n-4)(n-5)(n-6)(n-7)(n-8)(n-9)$$

$$(n^8 - 46n^7 + 815n^6 - 6191n^5 + 3639n^4 + 259545n^3 - 1691294n^2 + 3969851n - 2399936)p^{18}$$

$$\frac{1}{2} (n-2)(n-3)(n-4)(n-5)(n-6)(n-7)(n-8)(n-9)(n-10)(n^9 - 56n^8 + 1245n^7$$

$$- 12793n^6 + 33185n^5 + 566075n^4 - 6107264n^3 + 24180203n^2 - 34169983n - 5307479)p^{20}$$

Here is an instance for a graph of $n = 10^5$ nodes, a probability of $p = 5 \cdot 10^{-5}$, so that the mean node degree is about $\delta = 5$:

```
> seq(subs(n=10^5, p=5.0*10^(-5), [11, means(n, p, 11)]), 11=2..16);
[2, .3124843752 10^-7], [3, .7811562541 10^-6], [4, .00001952734406], [5, .0004881347839],
[6, .01220190522], [7, .3050049248], [8, 7.623903131], [9, 190.5632718], [10, 4763.129012],
[11, 119052.0291], [12, .2975586447 10^7], [13, .7437032079 10^8], [14, .1858737456 10^10],
[15, .4645449668 10^11], [16, .1160990805 10^13]
```

It appears that there is a sharp threshold between $l = 7$ and $l = 8$. This is to be compared with the number of nodes reachable from the root of a tree with node degree $\delta = 5$ in l steps: $5^7 = 78125$ is just below $10^5 = 100000$ while $5^8 = 390065$ exceeds 10^5

The probability threshold after which the mean number of unordered pairs exceeds 1 is given by

```
> p0:=proc(nn, l) local p; fsolve(subs(n=nn, means(n, p, l))=1, p, 0..1)
end;
p0 := proc(nn, l) local p; fsolve(subs(n=nn, means(n, p, l))=1, p, 0..1) end
```

And this quantity is best visualised in terms of the mean node degree:


```
> delta:=proc(n, p) (n-1)*p; end;
delta := proc(n, p) (n-1)*p end
```

For a graph of size $n = 10^5$, we have

```
> seq([11, delta(10^5, p0(10^5, 11))], 11=2..10);
[2, 376.0612495], [3, 52.10059411], [4, 19.39256537], [5, 10.71794899], [6, 7.218218497],
[7, 5.442530684], [8, 4.403820165], [9, 3.735053414], [10, 3.273942473]
```

The thresholds are fairly sharp as attested by plots of the mean values of the number of (unordered) avoiding pairs when the probability p increases (the horizontal axis gives the mean degree $\delta = (n-1)p$).

```
> nn:=10^5:
plot([seq(means(nn, delta/(nn-1), 11), 11=3..8)], delta=0..20, mu=0..1,
thickness=3);
```



(The curves from left to right correspond to path length $l = 8, 7, 6, 5, 4, 3$.)

Asymptotics. For dense graphs and large size, a numerical pattern clearly emerges. Here is what happens for mean degree $\delta = 10$ and size $n = 10^5$:

```

> seq(subs(n=10^5, p=1.0*10^(-4), [11, means(n, p, 11)]), 11=2..20);
[2, .4999750003 10^-6], [3, .00004999400026], [4, .004999000080], [5, .4998500187],
  [6, 49.97900378], [7, 4997.200688], [8, 499640.1157], [9, .4995501832 10^8],
  [10, .4994502767 10^10], [11, .4993404020 10^12], [12, .4992205655 10^14],
  [13, .4990907745 10^16], [14, .4989510366 10^18], [15, .4988013600 10^20],
  [16, .4986417537 10^22], [17, .4984722270 10^24], [18, .4982927901 10^26],
  [19, .4981034535 10^28], [20, .4979042286 10^30]

```

The mantissas all start with 0.49. This corresponds to a regime where the distance l is small so that only dominant terms in the mean value polynomial intervene. The mean number of avoiding pairs is then approximately

```

> mu=1/2*n^(2*l)*p^(2*l+2)*(1+o(1));

```

$$\mu = \frac{1}{2} n^{(2l)} p^{(2l+2)} (1 + o(1))$$

or, in terms of the mean degree parameter δ :

```

> mu=1/2*delta^(2*l)*p^2;

```

$$\mu = \frac{1}{2} \delta^{(2l)} p^2$$

This simplified formula explains the numerical regularities seen above, where the mean number increases by a factor of about 100 when l increases by 1.

Conversely, a clear example of **nonasymptotic regime** is provided by $p = 10^{\left(\frac{1}{8}\right)}$ and $n = 10^3$, that is $\delta = 1.333$.

```

> nn:=10^3; dd:=10.0^(1/8); pp:=dd/(nn-1);for ll from 10 to 28 by 2
do
l=ll,ExactMean=evalf(subs(n=nn,p=pp,means(n,p,ll)),5),ApproxMean=e
valf(1/2*nn^(2*ll)*pp^(2*ll+2),5) od;

nn := 1000
dd := 1.333521432
pp := .001334856288
l = 10, ExactMean = .00014443, ApproxMean = .00028743
l = 12, ExactMean = .00043780, ApproxMean = .00091255
l = 14, ExactMean = .0013163, ApproxMean = .0028974
l = 16, ExactMean = .0039256, ApproxMean = .0091990
l = 18, ExactMean = .011613, ApproxMean = .029207
l = 20, ExactMean = .034072, ApproxMean = .092730
l = 22, ExactMean = .099155, ApproxMean = .29442
l = 24, ExactMean = .28620, ApproxMean = .93475
l = 26, ExactMean = .81934, ApproxMean = 2.9678
l = 28, ExactMean = 2.3262, ApproxMean = 9.4225

```

The ratio between the exact formula and the simplified approximation is typically of about 1 to 2 or 1 to 4, with the approximation being systematically optimistic. An explanation is that, apart from asymptotic approximations in n , the simplified formula precisely neglects the exclusion effect on edges and this is permissible only in dense large graphs having high connectivity.

Conclusion

Permutations with constrained succession gaps are accessible to `combstruct` and `gfun`. This is achieved via the notion of templates that are nicely decomposable in conjunction with the inclusion-exclusion principle that is expressed by a simple integral transformation. Generating functions and recurrences result automatically and this leads to explicit binomial formulae. An application to robustness in a random interconnection graph derives from this approach and we have shown how to determine the mean number of edge-disjoint pairs between a source and a target in such a random graph.